Pensieve header: October 13: A Faster Jones Program.

Today. A faster Jones, then whatever you may suggest, then EIWL 9-12, then, if time, Patterns.

Topics (in no particular order). Whatever you may suggest; whatever comes to my mind; ~~the Fibonacci numbers~~; the Catalan numbers; ~~the Jones polynomial~~; a more efficient Jones algorithm; ~~a riddle on spheres~~; Khovanov homology; Γ-calculus; the Hopf fibration; Hilbert's 13th problem; non-commutative Gaussian elimination; free Lie algebras; the Baker-Campbell-Hausdorff formula; wacky numbers; an order 4 torus; the Schwarz Lantern; knot colourings; the Temperley-Lieb pairing; the dodecahedral link; sound experiments; barycentric subdivisions; a Peano curve; braid closures and Vogel's algorithm; the insolubility of the quintic; phase portraits; the Mandelbrot set; shadows of the Cantor aerogel; quilt plots; some image transformations; De Bruijn graphs; the Riemann series theorem; finite type invariants and the Willerton fish; the Hanoi towers.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
PD[Knot[3, 1]]
```

```
KnotTheory: Loading precomputed data in PD4Knots`.
```

```
PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]
```

```
Knot[8, 17]
```

```
Knot[8, 17]
```

```
Knot[8, 17] // PD
```

```
PD[X[6, 2, 7, 1], X[14, 8, 15, 7], X[8, 3, 9, 4], X[2, 13, 3, 14],
 X[12, 5, 13, 6], X[4, 9, 5, 10], X[16, 12, 1, 11], X[10, 16, 11, 15]]
```

```
Jones[PD[Knot[3, 1]]][q]
```

$$-\frac{1}{q^4}+\frac{1}{q^3}+\frac{1}{q}$$

```
AllKnots[{3, 10}] // Length
```

```
249
```

```
SetAttributes[P, Orderless];
JP[K_Times, opts___Rule] := Module[{verb, n, b1, b2, b3, b4, b5, w, J},
   verb = Verbose /. {opts} /. Verbose → False;
   n = Length[K];
   If[verb, Print["K has ", n, " crossings."]];
   b1 = K //. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + B P[j, k] P[i, l];
   b2 = Expand[b1];
   b3 = b2 //. P[i_, j_] P[j_, k_] :→ P[i, k];
   b4 = b3 //. {P[i_, j_]^2 → d, P[i_, i_] → d};
   b5 = Expand[b4 //. {B → 1 / A, d → -A^2 - 1 / A^2}];
   If[verb, Print["The Kauffman bracket is "]];
   If[verb, Print[b5]];
   w = K /. {Times → Plus, X[_, 1, _, 2 n] → 1,
       X[_, 2 n, _, 1] → -1, X[_, j_, _, l_] :→ If[j > l, 1, -1]};
   If[verb, Print["The writhe is "]];
   If[verb, Print[w]];
   If[verb, Print["The Jones Polynomial is "]];
   J = Expand@Cancel[((-A^3)^-w b5)/(-A^2 - 1 / A^2) /. A → q^(-1/4)]
  ];
JP[K_PD, opts___] := JP[Times @@ K, opts];
JP[K_Knot, opts___] := JP[PD@K, opts];
```

```
JP[Knot[3, 1], Verbose → True]
```

K has 3 crossings.

The Kauffman bracket is

$$\frac{1}{A^7} + \frac{1}{A^3} + A - A^9$$

The writhe is

-3

The Jones Polynomial is

$$-\frac{1}{q^4} + \frac{1}{q^3} + \frac{1}{q}$$

```
Timing[tab1 = Table[JP[K], {K, AllKnots[{3, 10}]}];]
```

{52.2031, Null}

```
tab2 = Table[Jones[K][q], {K, AllKnots[{3, 10}]}];
```

KnotTheory: Loading precomputed data in Jones4Knots`.

```
tab1 == tab2
```

True

```
Union[tab1] // Length
```

242

A 48-crossing knot

```
Import["http://drorbn.net/AcademicPensieve/2016-09/GST48-Marked.png"]
```



```
GST48 = PD[
  X[01, 15, 02, 14], X[29, 02, 30, 03],
  X[40, 04, 41, 03], X[04, 44, 05, 43], X[05, 26, 06, 27],
  X[95, 07, 96, 06], X[07, 01, 08, 96], X[08, 14, 09, 13],
  X[28, 09, 29, 10], X[41, 11, 42, 10],
  X[11, 43, 12, 42], X[12, 27, 13, 28], X[15, 31, 16, 30],
  X[61, 16, 62, 17], X[72, 17, 73, 18],
  X[83, 18, 84, 19], X[34, 20, 35, 19], X[20, 89, 21, 90],
  X[92, 21, 93, 22], X[22, 79, 23, 80],
  X[23, 68, 24, 69], X[24, 57, 25, 58], X[56, 25, 57, 26],
  X[31, 63, 32, 62], X[32, 74, 33, 73],
  X[33, 85, 34, 84], X[35, 50, 36, 51], X[81, 37, 82, 36],
  X[70, 38, 71, 37], X[59, 39, 60, 38],
  X[54, 39, 55, 40], X[55, 45, 56, 44], X[45, 59, 46, 58],
  X[46, 70, 47, 69], X[47, 81, 48, 80],
  X[91, 49, 92, 48], X[49, 91, 50, 90], X[82, 52, 83, 51],
  X[71, 53, 72, 52], X[60, 54, 61, 53],
  X[74, 63, 75, 64], X[85, 64, 86, 65], X[65, 76, 66, 77],
  X[66, 87, 67, 88], X[94, 67, 95, 68],
  X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79] ];
```

2^{48}

281 474 976 710 656

```
b1 = K //. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + B P[j, k] P[i, l];
b2 = Expand[b1];
b3 = b2 //. P[i_, j_] P[j_, k_] :→ P[i, k];
b4 = b3 //. {P[i_, j_]^2 → d, P[i_, i_] → d};
b5 = Expand[b4 //. {B → 1/A, d → -A^2 - 1/A^2}];
```

```
Knot[3, 1] // PD
```

PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]

```
J1 = X[1, 4, 2, 5] /. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + A^-1 P[j, k] P[i, l]
```

$\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]$

```
J2 = Expand[J1]
```

$\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]$

```
J3 = J2 //. P[i_, j_] P[j_, k_] :→ P[i, k]
```

$\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]$

```
J4 = J3 /. {P[i_, j_]^2 → -A^2 - 1/A^2, P[i_, i_] → -A^2 - 1/A^2}
```

$\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]$

```
J5 = Expand[J4]
```

$\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]$

```
J1 = J5 * (X[3, 6, 4, 1] /. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + A^-1 P[j, k] P[i, l])
```

$\left(\frac{P[1,5]\,P[2,4]}{A} + A\,P[1,4]\,P[2,5]\right)\left(A\,P[1,4]\,P[3,6] + \frac{P[1,3]\,P[4,6]}{A}\right)$

```
J2 = Expand[J1];
J3 = J2 //. P[i_, j_] P[j_, k_] :→ P[i, k];
J4 = J3 /. {P[i_, j_]^2 → -A^2 - 1/A^2, P[i_, i_] → -A^2 - 1/A^2};
J5 = Expand[J4]
```

$\frac{P[2,6]\,P[3,5]}{A^2} + P[2,5]\,P[3,6] - A^4\,P[2,5]\,P[3,6]$

```
J1 = J5 * (X[5, 2, 6, 3] /. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + A^-1 P[j, k] P[i, l])
```

$\left(\frac{P[2,6]\,P[3,5]}{A} + A\,P[2,5]\,P[3,6]\right)\left(\frac{P[2,6]\,P[3,5]}{A^2} + P[2,5]\,P[3,6] - A^4\,P[2,5]\,P[3,6]\right)$

```
J2 = Expand[J1];
J3 = J2 //. P[i_, j_] P[j_, k_] :> P[i, k];
J4 = J3 /. {P[i_, j_]^2 -> -A^2 - 1/A^2, P[i_, i_] -> -A^2 - 1/A^2};
J5 = Expand[J4]
```

$$\frac{1}{A^7} + \frac{1}{A^3} + A - A^9$$

```
PD[Knot[3, 1]]
```

PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]

```
FJ[K_PD] := Module[{J = 1, todo, front, v, x, n, w},
    n = Length[K];
    todo = List @@ K;
    front = {};
    v[x_X] := Length[front ∩ (List @@ x)];
    While[todo =!= {},
     x = RandomChoice@MaximalBy[todo, v];
     J = Expand[J * (x /. X[i_, j_, k_, l_] :> A P[i, j] P[k, l] + A^-1 P[j, k] P[i, l])];
     J = J //. P[i_, j_] P[j_, k_] :> P[i, k];
     J = Expand[J /. {P[i_, j_]^2 -> -A^2 - 1/A^2, P[i_, i_] -> -A^2 - 1/A^2}];
     todo = DeleteCases[todo, x];
     (* todo=Complement[todo, {x}]; *)
     (* front=Complement[front∪(List@@x), front∩(List@@x)]) *)
     front = front ∪ (List @@ x);
    ];
    w = K /. {PD -> Plus, X[_, 1, _, 2 n] -> 1,
       X[_, 2 n, _, 1] -> -1, X[_, j_, _, l_] :> If[j > l, 1, -1]};
    Expand@Cancel[(-A^3)^-w J / (-A^2 - 1/A^2) /. A -> q^(-1/4)]
  ];
FJ[K_Knot] := FJ[PD@K];
```

```
FJ[Knot[3, 1]]
```

$$-\frac{1}{q^4} + \frac{1}{q^3} + \frac{1}{q}$$

```
Timing[tab3 = FJ /@ AllKnots[{3, 10}];]
```

{2.20313, Null}

```
tab3 == tab1
```

True

```
FJ[GST48] // Timing
```

$$\left\{2.6875,\ 5 - \frac{1}{q^7} + \frac{1}{q^6} - \frac{1}{q^3} + \frac{3}{q^2} - \frac{3}{q} - 5 q + 5 q^2 - 3 q^3 - q^4 + 3 q^5 - 4 q^6 + 2 q^7 - q^8 + q^9 - q^{10} + q^{11} - q^{12} + 3 q^{13} - 4 q^{14} + 3 q^{15} - q^{16}\right\}$$

```
? MaximalBy
```

MaximalBy[{e_1, e_2, …}, f] returns a list of the e_i for which the value of $f[e_i]$ is maximal.
MaximalBy[{e_1, e_2, …}, f, n] returns a list of the e_i corresponding to the n largest $f[e_i]$.
MaximalBy[f] represents an operator form of MaximalBy that can be applied to an expression. >>

```
rands = RandomReal[{-1, 1}, 20]
```

```
{0.934427, -0.671289, 0.702115, -0.560976, 0.498471, -0.0442471,
 0.409253, 0.0841965, 0.715116, 0.158044, 0.148088, 0.914499, 0.63077,
 0.215494, 0.329929, 0.447944, -0.345088, -0.500468, -0.891825, 0.41459}
```

```
Max[rands]
```

```
0.934427
```

```
Min [rands]
```

```
-0.891825
```

```
-Max[-rands]
```

```
-0.891825
```

```
MaximalBy[rands, Abs]
```

```
{0.934427}
```

```
f[x_] := -x;
MaximalBy[rands, f]
```

```
{-0.891825}
```

$(f[x] + g)^3$ // Expand

$g^3 - 3 g^2 x + 3 g x^2 - x^3$

```
? Function
```

Function[$body$] or $body$& is a pure function. The formal parameters are # (or #1), #2, etc.
Function[x, $body$] is a pure function with a single formal parameter x.
Function[{x_1, x_2, …}, $body$] is a pure function with a list of formal parameters. >>

```
Function[-#][77]
```

```
-77
```

```
Function[#1 + #2][3, 4]
```

```
7
```

```
(#1 + #2) &[4, 5]
```

```
9
```

```
MaximalBy[rands, -# &]
```

```
{-0.891825}
```

```
Function[{x, y}, x + y][5, 6]
```

```
11
```

```
(x ↦ x^2)[5]
```

```
25
```

```
Clear[BFJ];
BFJ[K_PD] := Module[{J = 1, todo = List @@ K, ocean = {}, x, n = Length[K], w},
   While[todo =!= {},
    x = RandomChoice@MaximalBy[todo, x ↦ Length[ocean ⋂ (List @@ x)]];
    J = Expand[J * (x /. X[i_, j_, k_, l_] :→ A P[i, j] P[k, l] + A^-1 P[j, k] P[i, l])];
    J = J //. P[i_, j_] P[j_, k_] :→ P[i, k];
    J = Expand[J /. P[_, _]^2 | P[i_, i_] → -A^2 - 1 / A^2];
    todo = DeleteCases[todo, x];
    ocean = ocean ⋃ (List @@ x);
   ];
   w = Plus @@ K /.
     {X[_, 1, _, 2 n] → 1, X[_, 2 n, _, 1] → -1, X[_, j_, _, l_] :→ If[j > l, 1, -1]};
   Expand@Cancel[((-A^3)^-w J)/(-A^2 - 1 / A^2) /. A → q^(-1/4)]
  ];
BFJ[K_Knot] := BFJ[PD@K];
```

```
BFJ[Knot[3, 1]] == FJ[Knot[3, 1]]
```

```
True
```

```
Cases[{1, 2, 3, 3.14, 4, 5}, _Integer]
```

```
{1, 2, 3, 4, 5}
```

```
Cases[{1, 2, 2.78, 3, 3.14, 4, 5}, 3.14 | _Integer]
```

```
{1, 2, 3, 3.14, 4, 5}
```